

# Office VBA 中 ADO 错误处理研究

## Researching ADO Error Handler in Office VBA

彭江平 (湖南大学会计学院信息管理系 410079)

**摘要:** 针对 VBA 一般情况下不能处理 ADO 错误信息的问题, 分析了问题的形成机理, 并给出了一个完整的解决方案, 对在 VBA 或 VB 中处理 ADO 错误有一定的指导意义。

**关键词:** VBA ADO 错误处理

### 1 引言

VBA (Visual Basic for Application) 作为对 Microsoft Office 的二次开发的语言, 随着各行各业计算机应用的普及, 在许多比较复杂或专业的办公信息系统中得到大量的应用; ADO (ActiveX Data Object) 作为通用的数据访问对象, 在许多与数据库相关的信息系统的开发中得到广泛的应用。自 Microsoft Office 2000 以后, Office 也提供了对 ADO 的支持, 因此, 结合应用 VBA 与 ADO, 充分发挥它们各自的优势, 为 Office 应用与管理信息系统的结合提供了一条良好的途径。

最近笔者因参与设计一个基于 VBA 与 ADO 的信息系统的开发, 为设计一个比较通用的错误处理模块, 有机会对 VBA 中 ADO 错误信息的处理进行比较系统的分析, 参考了大量相关文献, 并得到了一个较好的解决方案, 期望能对正在从事类似工作的同行有一定的借鉴作用。

### 2 VBA 中 ADO 错误处理机理分析

#### 2.1 VBA 中一般的错误处理机理分析

设计良好的错误处理代码不论具体的实现方式如何, 也不论其代码是简单还是复杂, 总是

由下面三部分组成:

- (1) 说明当错误发生时做什么;
- (2) 处理发生的错误;
- (3) 说明程序如何继续执行。在 VBA 中, 语句 "On Error" 可以用来完成第一部分的功能, 而第二部分与第三部分就是具体的错误处理程序来完成, 它是整个错误处理过程的核心。在具体的错误处理代码中, 一般都是采用依据对 VBA 提供的错误 Err 集合对象进行处理的某种方式来实现: 或者采用多分支语句针对不同的错误进行处理, 或者采用 "On Error Resume Next" 语句忽略相应的错误。

#### 2.2 VBA 中 ADO 错误处理机理分析

由 ADO 的错误处理机制可知, 包含 ADO 对象的任何操作能依据不同的 ADO 数据提供者产生一个或多个错误, 而从 ADO 操作中产生的每一个错误都会向 ADO Connection 连接对象对应的 Errors 集合中添加一个 Error 对象。如果另外一个 ADO 对象也产生了一个或多个错误, 则 ADO Connection 对象的 Errors 集合将被清空, 然后再添加新产生的 Error 对象。当在 VBA 中使用 ADO 对象时, 如有任何 ADO 错误发生, 在 VBA 的 Err 对象中将包含 ADO Connection 连接对象


的 Errors 集合的第一个 Error 对象, 即在 VBA 的 Err 对象中将包含第一个 ADO 错误对象的错误号及相关的错误信息。

基于以上两点, 似乎与 VBA 的标准错误处理机制相同, 可以通过 VBA 的 Err 对象集合对 ADO 错误进行处理。但是, 不难通过编程验证: 当发生 ADO 错误时, 采用通常的模式, 依据 VBA 的 Err 集合中产生的错误对象进行处理时, 会发现错误号是一个绝对值相当大的负数, 而错误信息描述一般是与相应的 ADO 数据提供程序相关的信息。因为错误号无法在 VBA 或 ADO 错误处理文档中查找到, 很难在错误处理程序中对此作进一步的控制。使用 VBA 的 Err 对象的错误编号设计错误处理程序至少存在如下两方面的问题:

(1) 错误号如果直接在 VBA 的多分支语句中使用, 可能会出现溢出的现象;

(2) 错误号与 ADO 数据提供者 (相应数据库) 的错误信息编号不一致, 很难对错误信息进行进一步的控制。

对于第一个问题的出现, 是由于在 VBA 的错误处理机制中, 对于介于 0 到 65535 之间错误号在 VBA 语言级进行了处理, 在错误处理程序可以直接使用, 而对于这之外的错误 (一般是相应的对象产生的错误) 的错误号是一个绝对值相当大的负数, 在 VBA 的错误处理代码中, 可以将错误号与 VBA 定义的常数 "vbObjectError" (其值为 -2147221504, 即 &H80040000) 相减, 再依据转换后的结果对应的错误号处理, 来实



现对不同对象产生的错误进行控制。

对于第二个问题,虽然采用对第一个问题处理的同样的方法,能将错误号转换成VBA能处理的范围,但是转换后的错误号仍然与数据提供者相对的错误编码不一致。事实上,由前面的分析可知,由ADO Connection连接对象传送到VBA的Err对象中的错误号是第一个ADO错误号,并不是与数据提供者相对应的错误号。进一步分析ADO规范可以发现,规范中定义在Errors集合中的每个Error对象都包含一条基础的数据提供程序中的消息,ADO并不给出提供程序特定的诊断信息,这就解析了为什么错误号不与数据提供者的错误号对应,而错误信息与相应的数据提供者相关联。这样,为了获得与数据提供者相一致的错误号,不能直接使用VBA的Err对象或ADO的Error对象中提供的错误号,而应使用ADO的Error对象提供的另一个属性“NativeError”,该属性给出了与相应的数

据提供者相关联的错误号。

在获得了与数据提供者相关联的错误号以后,就可基于数据提供者的错误代码信息设计用户友好的ADO数据库错误处理程序。例如,可以针对不同的数据库系统,设计相应的错误信息编码表,在给出相应的错误号后,获得合适的用户定义的错误信息;或者直接使用数据库系统提供的错误信息代码表,如在SQL Server 2000数据库系统中,错误信息代码表为“master..dbo.system-sages”,在该表中,针对不同的语言集定义了各个错误号对应的错误信息,依据该表可以设计多语言的用户友好的错误处理程序。

### 3 VBA中ADO错误处理实现

上面分析了在VBA中ADO的错误处理机制,本节将以具体的实例给出VBA中ADO错误处理程序的一个示例,该实例由三个子过程组成,其中第一个子过程实现ADO连接,并

打开ADO记录集,而第二个子过程是在应用完成后对ADO对象的后期处理,这两个子过程共同使用第三个错误处理子过程。具体的实现代码如下:

```
Public cn As ADODB.Connection '定义应用中要使用的ADO连接对象与记录集全局对象

Public rs As ADODB.Recordset

Sub DataBaseInit() '数据连接初始化工作

On Error GoTo ErrorHandler

Dim myConstr As String

Dim AppIni As String

Dim sErrDesc As String

AppIni = "D:\cai\INI\"

Set cn = New ADODB.Connection

'OLEDB Provider Connection Setup

myConstr = "File Name=" & AppIni & "testfile.udl"

Connd = False
```

```

Process.Show
cn.ConnectionString = myConstr
cn.Open
If cn.State = adStateOpen Then
    Conned = True
    myConstr = "select * from reportlist"
    Set rs = New ADODB.Recordset
    rs.Open myConstr, cn
    '其他操作
End If
Exit Sub
ErrorHandler:
    myErrorHandler
End Sub
Sub DataBaseRelease() '数据连接完成后
    的后期处理工作
    On Error GoTo ErrorHandler
    If Not rs Is Nothing Then
        If rs.State = adStateOpen Then rs.Close
    End If
    Set rs = Nothing
    If Not cn Is Nothing Then
        If cn.State = adStateOpen Then cn.Close
    End If
    Set cn = Nothing
Exit Sub
ErrorHandler:
    myErrorHandler
End Sub
Sub myErrorHandler()
    If Err.Number > 0 And Err.Number < 65535
Then '有效的 VBA 错误号
        Select Case Err.Number
            Case 432
                MsgBox "OLE 自动化对象对应的文件未找到, 请检查文件目录或文件是否存在!",
                ,"文件不存在"

```

```

Case 3002
    MsgBox "不能读取文件, 请检查文件目录或文件是否存在!", ,"文件不存在"
Case 3004
    MsgBox "文件正在被打开编辑, 关闭正在编辑的文件后再处理!", ,"文件存在"
Case Else
    MsgBox "错误号: " & Err.Number & vbCrLf & "错误来源: " & Err.Source & vbCrLf & "错误描述: " & Err.Description, "错误信息"
End Select
Else
    If cn.Errors.Count > 0 Then 'ADO 错误
        For Each erCur In cn.Errors
            'erCur.NativeError
            sErrDesc = sErrDesc & Err.Number & "*" & erCur.Number & "错误号: " & erCur.NativeError & vbCrLf & "错误来源: " & erCur.Source & vbCrLf & "错误描述: " & erCur.Description & vbCrLf
        Next erCur
        MsgBox sErrDesc, "ADO OLEDB 错误"
    Else '其他对象错误
        Select Case Err.Number - vbObjectError
            Case -65533
                MsgBox "不能读取 UDL 数据库配置文件, 请检查文件目录或文件是否存在!",
                ,"UDL 文件不存在"
            Case Else
                MsgBox "错误号: " & Err.Number & vbCrLf & "错误来源: " & Err.Source & vbCrLf & "错误描述: " & Err.Description, "错误信息"
        End Select
    End If
End If

```

End Sub

代码的核心是第三个错误处理子过程。首先将 VBA 的 Err 对象的错误处理分为两大类, 即由 VBA 语言本身产生的错误及由其他对象产生的错误; 然后, 针对由其他对象产生的错误, 通过判断 ADO Connection 连接对象的错误数决定是否 ADO 错误; 如果是 ADO 错误, 则借助 ADO Connection 连接对象的 Errors 对象集合的 Error 对象的 "NativeError" 属性获得与数据提供者相一致的错误号, 再在此基础上设计与数据提供者相关联的错误处理 (这样仅提供了最简单的错误信息提示处理, 实际使用时可以依据获得的与数据提供者相关联的错误号, 进行更深入的与数据提供者相关联的错误处理); 如果是由其他对象产生的错误, 则将错误号通过常数 "vbObjectError" 进行转换, 并进行相应的错误处理。

#### 4 结论

错误处理作为设计良好的信息系统中不可缺少的重要组成部分, 探讨其设计方法具有重要的理论意义。ADO 作为通用的数据访问组件, 可用于 Visual Basic、Visual C++、Delphi 与 VBA。本文讨论的内容不仅给出了在 VBA 中 ADO 错误处理的实现方法, 而且对在其他环境中 ADO 错误信息的处理也有一定的参考价值。■

#### 参考文献

- 1 晶宸工作室编著, Excel 2000 VBA 开发实例指南, 电子工业出版社, 2000.8.
- 2 刘曜, 在 Excel 上设置自定义函数定义报表, 计算机系统应用, 2002.6, P71-73.
- 3 三味工作室, SQL 外 Server 7.0 管理指南, 中国水利水电出版社, 1999.8.
- 4 王能斌, 数据库系统原理, 电子工业出版社, 2000.1.